# Maintainer: Alexey Pavlov <alexpux@gmail.com>

pkgname=setconf
pkgver=0.7.5
pkgrel=1
pkgdesc='Utility to easily change settings in configuration files'
arch=('any')
url='http://setconf.roboticoverlords.org/'
license=('GPL2')
depends=('python2')
source=("http://setconf.roboticoverlords.org/$pkgname-$pkgver.tar.xz")
sha256sums=('3aa5ded4c3e1996a78cbbcbd7b9af051e3a197f6e79fced9ebb118da902a52fe')

prepare() {
  sed -i 's:usr/bin/python:usr/bin/python2:' "$pkgname-$pkgver/$pkgname.py"
}

package() {
  cd "$pkgname-$pkgver"

  install -Dm755 "$pkgname.py" "$pkgdir/usr/bin/$pkgname"
  install -Dm644 "$pkgname.1.gz" "$pkgdir/usr/share/man/man1/$pkgname.1.gz"
}
